Numer. Math. 12, 231—241 (1968)

Handbook Series Linear Algebra

Tridiagonalization of a Symmetric Band Matrix*

Contributed by

H. R. SCHWARZ

Theoretical Background

The well known method proposed by GIVENS [1] reduces a full symmetric matrix $A=(a_{ik})$ of order n by a sequence of appropriately chosen elementary orthogonal transformations (in the following called Jacobi rotations) to tridiagonal form. This is achieved by $(n-1)(n-2)/2$ Jacobi rotations, each of which annihilates one of the elements a_{ik} with $|i-k|>2$. If this process is applied in one of its usual ways to a symmetric band matrix $A=(a_{ik})$ of order n and with the band width $m>1$, i.e. with

$$a_{ik}=0 \quad \text{for all } i \text{ and } k \text{ with } \quad |i-k|>m, \tag{1}$$

it would of course produce a tridiagonal matrix, too. But the rotations generate immediately nonvanishing elements outside the original band that show the tendency to fill out the matrix. Thus it seems that little profit with respect to computational and storage requirements may be taken from the property of the given matrix A to be of band type.

However, by a generalization of an idea of RUTISHAUSER [2], it is indeed possible to reduce a symmetric band matrix to tridiagonal form by a suitable sequence of Jacobi rotations while preserving the band property of the given matrix throughout the whole process (see [3]). A single step consists of the well known similarity transformation

$$A_{k+1}=U_k^T A_k U_k \quad (A_0=\text{given matrix } A), \tag{2}$$

where $U_k=U_k(p, q, \varphi)$ is an orthogonal matrix which deviates from the unit matrix only in the elements

$$u_{pp}=u_{qq}=cos(\varphi), \qquad u_{pq}=-u_{qp}=sin(\varphi) \quad \text{with} \quad p<q. \tag{3}$$

Herein φ denotes the real angle of rotation in the (p, q)-coordinate plane. The effect of the transformation (2) on A_k is a linear combination of the rows and columns with indices p and q, while all other elements of A_k remain unchanged.

* *Editor's note.* In this fascicle, prepublication of algorithms from the Linear Algebra Series of the Handbook for Automatic Computation is continued. Algorithms are published in ALGOL 60 reference language as approved by the IFIP. Contributions in this series should be styled after the most recently published ones.

The basic idea of the transformation is the following: The elimination of an element within the band of the matrix by an appropriate Jacobi rotation generates just two identical, in general nonzero elements g in symmetric positions outside the band. To maintain the band type of the matrix these elements are eliminated by a sequence of additional rotations with the effect of shifting them downwards over m rows and m columns and finally beyond the border of the matrix.

For $n=10$, $m=3$ the complete process for the elimination of the element a_{14} is illustrated in (4), where for the reason of symmetry only the elements in and above the main diagonal are considered.

(4)



A first rotation with $U_1(3, 4, \varphi_1)$ and appropriate chosen angle φ_1 eliminates a_{14}, but generates an element g in the third row and seventh column. This element g is annihilated by a second rotation with $U_2(6, 7, \varphi_2)$ which on its turn produces another element g in the sixth row and the tenth column. A third rotation with $U_3(9, 10, \varphi_3)$ completes the elimination of a_{14}.

By analogy the element a_{13} is eliminated next by starting with a rotation defined by $U_4(2, 3, \varphi_4)$. Two additional rotations will sweep the appearing element g in the second row and sixth column downwards and over the border of the matrix.

So far the first row (and column) has been brought to the desired form. It is almost obvious that continuation of the process with the following rows will never destroy the already generated zero elements within the band and will terminate with the transformed tridiagonal matrix.

The sequence of Jacobi rotations for the complete transformation of a band matrix A of order n and band width $m>1$ is fully described in Table 1[1].

[1] A different strategy for the transformation of a band matrix to tridiagonal form consists of a systematic reduction of the band width of the matrix (see [4, 5]). However, the present strategy is less time-consuming, since it requires a smaller number of Jacobi rotations and a marginally smaller number of multiplications. A similar reduction to that given here is achieved by the Householder transformation [2]. On the whole the extra complexity of the organization of HOUSEHOLDER in this case means that it does not have its usual superiority over GIVENS.

Table 1. *Sequence of rotations*

Elimination of the elements in row j	First rotation	Position of the first element g (if any)	Additional rotations (if necessary)
$a_{j,j+k}$ ($k=m, m-1, \ldots, 2$) only if $j+k \leqq n$	$U(j+k-1, j+k, \varphi)$	$(j+k-1, j+k+m)$ only if $j+k+m \leqq n$	$U(j+k+\mu m-1, j+k+\mu m, \varphi)$ $\left(\mu=1, 2, \ldots, \left[\frac{n-k-j}{m}\right]\right)$
$j=1, 2, \ldots, n-2$			

It should be noted that for all rotations the two indices p and q differ by one, that is the corresponding transformations change in all cases two adjacent rows and columns. Furthermore the first rotation is just an extrapolated case of the additional rotations. This simplifies the procedure. There exists only a slight difference in the determination of the angles φ. For the first rotation the angle φ is determined from the equations

$$a_{j,j+k} \cos\varphi + a_{j,j+k-1} \sin\varphi = 0, \quad \cos^2\varphi + \sin^2\varphi = 1, \tag{5}$$

whereas for the additional rotations $(\mu = 1, 2, \ldots, [(n-k-j)/m])$ the equations

$$g \cdot \cos\varphi + a_{j+k+(\mu-1)m-1, j+k+\mu m-1} \sin\varphi = 0, \quad \cos^2\varphi + \sin^2\varphi = 1 \tag{6}$$

are to be used.

The complete method as described performs the operation

$$A \to J = V^T A V \tag{7}$$

in a finite number of steps N_{rot}, where J is tridiagonal and V is orthogonal. V is the product of all Jacobi rotations which were used for achieving the transformation (7), i.e.

$$V = U_1 U_2 U_3 \ldots U_{N_{\text{rot}}}. \tag{8}$$

Applicability

The present algorithm may be applied to any symmetric band matrix A of order n and band width $m > 1$ to reduce it to a tridiagonal matrix by a finite sequence of Jacobi rotations. The method is especially advantageous if $m \ll n$.

The application of the process is to be recommended, if *all* the eigenvalues of a symmetric band matrix are to be computed, since the combination of this bandreduction with the QD-algorithm shows a remarkable saving of computing time in comparison with the determination of all eigenvalues by the LR-transformation. The relation of the corresponding amount of computing time is approximately $1:m$.

In addition, the present procedure gives the connection to the save method of bisection for computing the eigenvalues in a given interval or the eigenvalues with a prescribed index [6], or for computing the eigenvalues by the rational QR-transformation [7].

Formal Parameter List

Input to procedure *bandrd*

n order of the matrix A (and of the matrix V).

m band width of the matrix A (= number of diagonals above the main diagonal).

matv boolean variable to decide whether the transformation matrix V (8) is to be computed (*matv* = **true**) or not (*matv* = **false**).

a **array** $a[1:n, 0:m]$ are the elements of the given band matrix A arranged in a rectangular array as described in "Organizational and Notational Details". The elements $a[i, k]$ with $i+k>n$ are irrelevant.

Output of procedure *bandrd*

a the elements of the resulting tridiagonal matrix J (7). The diagonal elements of J are given by $a[i, 0]$ $(i=1, 2, \ldots, n)$, the elements of the superdiagonal are given by $a[i, 1]$ $(i=1, 2, \ldots, n-1)$. The other elements $a[i, 2], a[i, 3], \ldots, a[i, m]$ are meaningless.

v if *matv* = **true**, the **array** $v[1:n, 1:n]$ contains the elements of the orthogonal transformation matrix V (8). This is a full square matrix. If *matv* = **false**, no values are assigned to $v[i, k]$.

ALGOL Procedure

```
procedure bandrd (n, m, matv) trans: (a) res: (v);
    value n, m, matv;
    integer n, m; boolean matv; array a, v;
    comment Transforms a real and symmetric band matrix A of order n
            and band width m (i.e. with a[i, k] = 0 for all i and k with
            abs(i - k) > m) to tridiagonal form by an appropriate finite
            sequence of Jacobi rotations. The elements of A in and above
            the main diagonal are supposed as a rectangular array (array
            a[1:n, 0:m]), where a[i, j] denotes the element in the i-th
            row and (i + j)-th column of the usual representation of a
            matrix. During the transformation the property of the band
            matrix is maintained. The method yields the tridiagonal
            matrix J = (VT)AV (where VT is the transpose of V), the ele-
            ments of which are a[i, 0] (i = 1(1)n) and a[i, 1] (i = 1(1)n-1).
            If the parameter matv is given the value true, the orthogonal
            matrix V is delivered as array v[1:n, 1:n], otherwise not;
  begin
    integer j, k, l, r, maxr, maxl, ugl;
    real b, c, s, c2, s2, cs, u, u1, g;
    if matv then
        for j := 1 step 1 until n do
            for k := 1 step 1 until n do
                v[j, k] := if j = k then 1 else 0;
    for k := 1 step 1 until n - 2 do
```

```
begin
   maxr := if n − k < m then n − k else m;
   for r := maxr step −1 until 2 do
   begin
      for j := k + r step m until n do
      begin
         if j = k + r then
         begin
            if a[k, r] = 0 then goto endr;
            b := −a[k, r − 1]/a[k, r];
            ugl := k
         end
         else
            begin
               if g = 0 then goto endr;
               b := −a[j − m − 1, m]/g;
               ugl := j − m
            end;
         s := 1/sqrt(1 + b↑2); c := b × s;
         c2 := c↑2; s2 := s↑2; cs := c × s;
         u := c2 × a[j − 1, 0] − 2 × cs × a[j − 1, 1] + s2 × a[j, 0];
         u1 := s2 × a[j − 1, 0] + 2 × cs × a[j − 1, 1] + c2 × a[j, 0];
         a[j − 1, 1] := cs × (a[j − 1, 0] − a[j, 0]) + (c2 − s2) × a[j − 1, 1];
         a[j − 1, 0] := u;
         a[j, 0] := u1;
         for l := ugl step 1 until j − 2 do
         begin
            u := c × a[l, j − l − 1] − s × a[l, j − l];
            a[l, j − l] := s × a[l, j − l − 1] + c × a[l, j − l];
            a[l, j − l − 1] := u
         end l;
         if j ≠ k + r then
            a[j − m − 1, m] := c × a[j − m − 1, m] − s × g;
         maxl := if n − j < m − 1 then n − j else m − 1;
         for l := 1 step 1 until maxl do
         begin
            u       := c × a[j − 1, l + 1] − s × a[j, l];
            a[j, l] := s × a[j − 1, l + 1] + c × a[j, l];
            a[j − 1, l + 1] := u
         end l;
         if j + m ≤ n then
         begin
            g := −s × a[j, m];
            a[j, m] := c × a[j, m]
         end;
         if matv then
            for l := 1 step 1 until n do
```

```
                    begin
                       u       := c×v[l, j−1] − s×v[l, j];
                       v[l, j] := s×v[l, j−1] + c×v[l, j];
                       v[l, j−1] := u
                    end l;
                 end j;
endr:         end r
           end k;
           a[n, 1] := 0
       end bandrd;
```

Organizational and Notational Details

Since symmetry of the given matrix A is preserved throughout the process, it is sufficient to know and to work with the elements of A in and above the main diagonal. The band matrix A is stored as an **array** $a[1:n, 0:m]$, where $a[i, j]$ $(i=1, 2, \ldots, n; j=0, 1, \ldots, m)$ denotes the element in the i-th row and $(i+j)$-th column of A in its usual arrangement. Thus $a[i, 0]$ is the i-th element in the diagonal, $a[i, 1]$ the element in the first superdiagonal of the i-th row, and so on. For example, the matrix

$$
A=\begin{bmatrix}
5 & -4 & 1 & & & & \\
-4 & 6 & -4 & 1 & & & \\
1 & -4 & 6 & -4 & 1 & & \\
 & 1 & -4 & 6 & -4 & 1 & \\
 & & 1 & -4 & 6 & -4 & 1 \\
 & & & 1 & -4 & 6 & -4 \\
 & & & & 1 & -4 & 5
\end{bmatrix} \tag{9}
$$

of order $n=7$ with $m=2$ is stored as

$$
\boxed{\begin{array}{ccc}
5 & -4 & 1 \\
6 & -4 & 1 \\
6 & -4 & 1 \\
6 & -4 & 1 \\
6 & -4 & 1 \\
6 & -4 & \times \\
5 & \times & \times
\end{array}} \tag{10}
$$

The elements in the lower right triangle marked by an $\times$ are irrelevant, since they are not used in the procedure *bandrd*.

The resulting tridiagonal matrix $J=V^T A V$ is contained in the same array a, i.e. the diagonal elements of J are given by $a[i, 0]$, the superdiagonal elements by $a[i, 1]$. For convenience the value of $a[n, 1]$ (that lies actually outside the matrix) is set equal to zero.

If required ($matv$ = **true**), the transformation matrix V is delivered as **array** $v[1:n, 1:n]$. If on the other hand V is not of interest ($matv$ = **false**), the actual counterpart of v can be any declared twodimensional array, thus e.g. a call like *bandrd* (70, 5, **false**, a, a) is meaningful.

The Jacobi rotations with $j-1$ and j as rotation indices and $c = \cos\varphi$, $s = \sin\varphi$ as determined from (5) or (6) are performed in three parts:

a) The new elements that are transformed by both the row and the column combinations are computed according to the usual formulas.

b) The new elements above the diagonal, undergoing only the column combination, are formed. Distinction has to be made whether the rotation annihilates an element within or the element g outside the band of the matrix.

c) The new elements to the right of the diagonal, transformed only by the row combination, are computed. Two cases have to be distinguished whether the rotation produces a new element g outside the band or not.

Numerical Properties

For the total number of rotations N_{rot} of the process an upper bound may be derived by neglecting the effects at the lower end of the band:

$$N_{\mathrm{rot}} \leqq n^2 \cdot (m-1)/(2m). \tag{11}$$

Each rotation involves a square root and in the normal case $8m+13$ multiplications. The total number of multiplications needed for the complete transformation is therefore bounded by the expression

$$N_{\mathrm{mult}} \leqq n^2 (m-1)(4 + 6.5/m). \tag{12}$$

For $m \ll n$ the method is indeed an n^2-process.

According to WILKINSON [5] we have the approximate error bound

$$\left[\frac{\Sigma(\mu_i-\lambda_i)^2}{\Sigma\lambda_i^2}\right]^{\frac{1}{2}} \leqq 12 \cdot \Theta \cdot n^{1.5}(1+6\Theta)^{4n-7} \cdot \frac{m-1}{m}, \tag{13}$$

where λ_i are the eigenvalues of the given matrix A, μ_i are the eigenvalues of the computed tridiagonal matrix J and Θ is the smallest positive number such that in the computer $1+\Theta \neq 1$.

Testresults

The procedure *bandrd* has been tested on the computer CDC 1604-A* of the Swiss Federal Institute of Technology, Zurich for the following examples:

a) The matrix (9) has been transformed to tridiagonal form J, and the eigenvalues of J have been computed by the method of bisection. The elements of J as well as the eigenvalues μ_i of J together with the eigenvalues λ_i of A as determined by the method of JACOBI [8] are listed in Table 2.

* The CDC 1604-A has a 36-bit mantissa and a binary exponent ranging from −1024 to +1023.

Table 2

i	$j\,[i,i]$	$j\,[i,i+1]$	$\mu\,[i]$	$\lambda\,[i]$
1	5.0000000000	−4.1231056257	0.0231 7730229	0.0231 7730233
2	7.8823529418	−4.0348825039	0.34314575048	0.34314575054
3	7.9535662945	−4.0166055335	1.5243189787	1.5243189787
4	7.9748041817	−3.9975334001	4.0000000001	3.9999999999
5	7.6058064229	−2.9758282822	7.6472538968	7.6472538965
6	3.3461613147	−0.4538408623	11.656854250	11.656854250
7	0.2373088458	0	14.805249823	14.805249823

b) The matrix $B=8C-5C^2+C^3$ of order 44, where C denotes the tridiagonal matrix with elements $c[i,i]=2$, $c[i,i+1]=1$, i.e.

$$(14)\qquad B=\begin{bmatrix}
5 & 2 & 1 & 1 & & & & & & & \\
2 & 6 & 3 & 1 & 1 & & & & & & \\
1 & 3 & 6 & 3 & 1 & 1 & & & & & \\
1 & 1 & 3 & 6 & 3 & 1 & 1 & & & & \\
 & 1 & 1 & 3 & 6 & 3 & 1 & 1 & & & \\
 & & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & & \\
 & & & & & & 1 & 1 & 3 & 6 & 2 \\
 & & & & & & & 1 & 1 & 2 & 5
\end{bmatrix}$$

has been used to determine its eigenvalues. The elements of the transformed tridiagonal matrix J are listed in Table 3 together with the computed eigenvalues μ_i of J by the method of bisection as well as the exact eigenvalues λ_i of B with

$$(15)\qquad \lambda_i=s_i^3-5s_i^2+8s_i,\qquad s_i=4\cdot\sin^2(2i^0)\qquad (i=1,2,\ldots,44).$$

c) The matrix A of order $n=30$ with $m=3$

$$(16)\qquad A=\begin{bmatrix}
10 & 1 & 1 & 1 & & & & & & & & \\
1 & 10 & 0 & 0 & 1 & & & & & & & \\
1 & 0 & 10 & 0 & 0 & 1 & & & & & & \\
1 & 0 & 0 & 9 & 0 & 0 & 1 & & & & & \\
 & 1 & 0 & 0 & 9 & 0 & 0 & 1 & & & & \\
 & & 1 & 0 & 0 & 9 & 0 & 0 & 1 & & & \\
 & & & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & & \\
 & & & & & 1 & 0 & 0 & 2 & 0 & 0 & 1 \\
 & & & & & & 1 & 0 & 0 & 1 & 0 & 0 \\
 & & & & & & & 1 & 0 & 0 & 1 & 0 \\
 & & & & & & & & 1 & 0 & 0 & 1
\end{bmatrix}$$

has been used to compute all its eigenvalues. The method of the band reduction yields the matrix J of Table 4. The subsequent application of the method of bisection delivers the eigenvalues μ_i of J. For comparison the eigenvalues λ_i of A have been determined by the Jacobi algorithm [8]. Although some of the

Table 3

i	$j[i, i]$	$j[i, i+1]$	$\mu[i]$	$\lambda[i]$
1	5.0000000000	2.4494897427	0.0388566346 2	0.0388566344 6
2	9.6666666667	3.7043517953	0.1538240643 2	0.1538240641 4
3	7.5114709853	4.4182460949	0.3401713221 8	0.3401713221 3
4	7.6884525213	3.8121582980	0.5902648041 4	0.5902648039 7
5	8.5830360828	3.8635865610	0.8939331615 3	0.8939331615 2
6	7.7597731312	4.2264403956	1.238954478 1	1.238954478 1
7	7.8107358423	3.8888572252	1.611643323 3	1.611643323 2
8	8.3723450440	3.9142790950	1.997510643 2	1.997510643 2
9	7.8258325770	4.1582582958	2.381966011 4	2.381966011 3
10	7.8711218853	3.9164482712	2.751029696 8	2.751029696 6
11	8.2749744193	3.9408159364	3.092021404 2	3.092021404 1
12	7.8578585475	4.1212019076	3.394193424 3	3.394193424 1
13	7.9082005882	3.9312475121	3.649278272 3	3.649278272 1
14	8.2159441281	3.9577102096	3.851924593 9	3.851924593 8
15	7.8781074404	4.0968242675	4.000000000 1	4.000000000 0
16	7.9337598720	−3.9410934000	4.004531845 4	4.004531845 8
17	8.1746472896	3.9695920067	4.005211952 8	4.005211953 2
18	7.8932128858	4.0789034129	4.034568007 3	4.034568007 7
19	7.9523352541	3.9480048752	4.052841589 4	4.052841589 4
20	8.1365859494	3.9651689592	4.078725692 3	4.078725692 4
21	7.8178365964	3.9479925008	4.094745370 3	4.094745370 0
22	7.4628371074	3.4781440573	4.120834653 0	4.120834653 4
23	6.6175117962	2.9637963604	4.140771753 8	4.140771754 1
24	5.2534192105	−2.4517781222	4.145898033 4	4.145898033 8
25	4.5422812540	2.0208165905	4.162503824 2	4.162503824 4
26	3.7256107620	1.7905673660	4.347374975 2	4.347374975 2
27	3.4574191781	1.5646029996	4.618033988 7	4.618033988 7
28	3.0729391709	−1.3383315621	4.981869044 4	4.981869044 5
29	3.1365536812	1.1419380160	5.442608728 4	5.442608728 6
30	3.2563743813	0.8506159836	6.000000000 0	6.000000000 0
31	3.4230726357	0.6344781230	6.649649032 5	6.649649032 5
32	3.6359837740	−0.4394343961	7.383034145 3	7.383034145 5
33	3.7783048010	0.3049897050	8.187692722 5	8.187692722 9
34	3.9169434055	0.1636998047	9.047576582 3	9.047576582 7
35	3.9304903009	0.0759018266	9.943563014 0	9.943563014 0
36	4.0638930104	−0.0428400764	10.85410196 6	10.85410196 6
37	4.0911967582	−0.0405738637	11.75597394 4	11.75597394 4
38	4.0900741534	−0.0336363562	12.62512829 1	12.62512829 1
39	4.0790565672	−0.0435345313	13.43756794 7	13.43756794 7
40	4.0615724516	0.0363513025	14.17024461 0	14.17024461 1
41	4.0592745087	−0.0318706517	14.80192758 0	14.80192758 1
42	4.0946552432	−0.0312974483	15.31401050 8	15.31401050 8
43	4.1121615408	−0.0084580378	15.69122271 9	15.69122271 9
44	4.0055665978	0	15.92221564 0	15.92221564 1

eigenvalues of A are almost triple, the transformed tridiagonal matrix J does not even nearly decompose as one might expect.

Table 4

i	$j[i, i]$	$j[i, i+1]$	$\mu[i]$	$\lambda[i]$
1	10.0000000000	1.732050807 6	0.2538058171	0.2538058170
2	9.6666666667	1.1055415967	0.2538058171	0.2538058171
3	8.7878787883	1.2694763675	0.2538058173	0.2538058171
4	8.0993006993	1.3870918026	1.7893213523	1.7893213522
5	7.3649630481	1.4295019112	1.7893213527	1.7893213527
6	6.6161666284	1.6030841144	1.7893213531	1.7893213529
7	6.5076783627	2.0653254573	2.9610588361	2.9610588358
8	6.2709028289	1.8863971727	2.9610588841	2.9610588842
9	4.6403533769	1.5370064289	2.9610589165	2.9610589161
10	3.5244488962	−1.7523840047	3.9960456418	3.9960456418
11	4.3171991227	2.9080849479	3.9960482020	3.9960482014
12	4.6866497255	1.8471929353	3.9960497561	3.9960497565
13	2.5184783050	0.2786808032	4.9996895667	4.9996895662
14	6.9959884597	−0.0512104592	4.9997824781	4.9997824777
15	6.0011669284	−0.0463833001	4.9998325860	4.9998325857
16	5.0562247743	−0.5697872379	5.9978378497	5.9978378494
17	10.640540849	0.5706708476	6.0002175230	6.0002175223
18	4.0431585993	0.0701584281	6.0012633582	6.0012633582
19	3.4977933009	1.7549123478	6.9637264189	6.9637264186
20	8.6523751242	−0.3854162757	7.0039517991	7.0039517985
21	1.2007407688	0.8906616138	7.0163502621	7.0163502620
22	2.8056476826	−3.1493000913	7.7615665168	7.7615665166
23	6.1014455162	0.0083686212	8.0389411172	8.0389411158
24	7.0033308959	0.02586397549	8.1138105888	8.1138105886
25	5.9994477055	0.03794630954	8.6878418569	8.6878418564
26	4.9947088899	−0.08112648663	9.2106786487	9.2106786475
27	3.9754502629	0.17699886640	9.4734106881	9.4734106874
28	2.5233531950	1.0249844551	10.174892584	10.174892583
29	0.7188981895	0.02656615397	10.746194184	10.746194183
30	1.7890424196	0	11.809310237	11.809310238

A second test has been performed at the National Physical Laboratory, Teddington. The results have been confirmed, but J. H. Wilkinson points out that although the algorithm is stable, the elements of the tridiagonal matrix are not always determined in a stable way. In the last example the later elements disagreed entirely (i.e. in the most significant figure) from those given in Table 4. But the eigenvalues, of course, agreed almost exactly as is proved by backward error analysis. This disagreement usually occurs with multiple or very close eigenvalues.

References

1. Givens, W.: A method for computing eigenvalues and eigenvectors suggested by classical results on symmetric matrices. Nat. Bur. Standards Appl. Math. Ser. **29**, 117—122 (1953).
2. Rutishauser, H.: On Jacobi rotation patterns. Proceedings of Symposia in Applied Mathematics, Vol. 15 Experimental Arithmetic, High Speed Computing and Mathematics, 1963, 219—239.
3. Schwarz, H. R.: Die Reduktion einer symmetrischen Bandmatrix auf tridiagonale Form. Z. f. angew. Math. u. Mech. (Sonderheft) **45**, T76—T77 (1965).

4. Schwarz, H. R.: Reduction of a symmetric bandmatrix to triple diagonal form. Commun. ACM **6**, 315—316 (1963).
5. Wilkinson, J. H.: The algebraic eigenvalue problem, 662 p. Oxford: Clarendon Press 1965.
6. Barth, W., R. S. Martin, and J. H. Wilkinson: Calculation of the eigenvalues of a symmetric tridiagonal matrix by the method of bisection. Numer. Math. **9**, 386—393 (1967).
7. Reinsch, C., and F. L. Bauer: Rational QR transformation with Newton shift for symmetric tridiagonal matrices. Numer. Math. **11**, 264—272 (1968).
8. Rutishauser, H.: The Jacobi method for real symmetric matrices. Numer. Math. **9**, 1—10 (1966).

Dr. H. R. Schwarz
Institut für angew. Mathematik
Eidg. Technische Hochschule
CH 8006 Zürich, Schweiz